

#delimit — Change delimiter

[Syntax](#)[Description](#)[Remarks and examples](#)[Also see](#)

Syntax

```
#delimit { cr | ; }
```

Description

The `#delimit` command resets the character that marks the end of a command. It can be used only in do-files or ado-files.

Remarks and examples

[stata.com](#)

`#delimit` (pronounced *pound-delimit*) is a Stata preprocessor command. `#commands` do not generate a return code, nor do they generate ordinary Stata errors. The only error message associated with `#commands` is “unrecognized `#command`”.

Commands given from the console are always executed when you press the *Enter*, or *Return*, key. `#delimit` cannot be used interactively, so you cannot change Stata’s interactive behavior.

Commands in a do-file, however, may be delimited with a carriage return or a semicolon. When a do-file begins, the delimiter is a carriage return. The command `'#delimit ;'` changes the delimiter to a semicolon. To restore the carriage return delimiter inside a file, use `#delimit cr`.

When a do-file begins execution, the delimiter is automatically set to carriage return, even if it was called from another do-file that set the delimiter to semicolon. Also, the current do-file need not worry about restoring the delimiter to what it was because Stata will do that automatically.

▷ Example 1

```

/*
    When the do-file begins, the delimiter is carriage return:
*/
use basedata, clear
/*
    The last command loaded our data.
    Let's now change the delimiter:
*/
#delimit ;
summarize sex
        salary ;
/*
    Because the delimiter is semicolon, it does not matter that our
    command took two lines.
    We can change the delimiter back:
*/

```

```
#delimit cr
summarize sex salary
/*
    Now our lines once again end on return.  The semicolon delimiter
    is often used when loading programs:
*/
capture program drop fix
program fix
    confirm var '1'
    #delimit ;
    replace '1' = . if salary>=. | salary==0 |
                    hours>=. | hours==0 ;
    #delimit cr
end
fix var1
fix var2
```



□ Technical note

Just because you have long lines does not mean that you must change the delimiter to semicolon. Stata does not care that the line is long. There are also other ways to indicate that more than one physical line is one logical line. One popular choice is `///`:

```
replace '1' = . if salary>=. | salary==0 | ///
                    hours>=. | hours==0
```

See [\[P\] comments](#).



Also see

[\[U\] 16.1.3 Long lines in do-files](#)

[\[U\] 18.11.2 Comments and long lines in ado-files](#)

[\[P\] comments](#) — Add comments to programs